
Natural Statements
This section describes the Natural statements required to process information between the mainframe and the PC
using Natural Connection.

These statements can be divided into the following groups:  

Transfer Data 

DOWNLOAD PC FILE (synonym for WRITE WORK FILE) 
UPLOAD PC FILE (synonym for READ WORK FILE) 

Download Reports 

DISPLAY 
PRINT 
WRITE 

Close a PC File 

CLOSE PC FILE (synonym for CLOSE WORK FILE) 

The above statements are listed in alphabetical order below. For detailed information, see the Natural Statements 
documentation.

CLOSE PC FILE

Function 

This statement is used to close a specific PC work file. It allows you to explicitly specify in a program that a PC
work file is to be closed.

A work file is also closed automatically when command mode is reached.

The settings in the NTWORK macro apply.

work-file-number 

The work-file-number is the number of the PC work file to be closed. This number must correspond to one of the 
work file numbers for the PC as defined to Natural.

Related Statements 

DOWNLOAD PC FILE, UPLOAD PC FILE.

Example

The following program demonstrates the use of the CLOSE PC FILE statement.

/* CLOSEEX: Example for CLOSE PC FILE
/*
DEFINE DATA LOCAL
  01 W-DAT   (A40)
  01 REC-NUM (N3)
  01 I       (P3)
END-DEFINE
*
REPEAT
  UPLOAD PC FILE 7 ONCE W-DAT                /* Data upload
  AT END OF FILE
    ESCAPE BOTTOM
  END-ENDFILE
  INPUT ’Processing file’ W-DAT (AD=O)
    /   ’Enter record number to display’ REC-NUM
  IF REC-NUM = 0
    STOP
  END-IF
  FOR I = 1 TO REC-NUM
    UPLOAD PC FILE 7 ONCE W-DAT
    AT END OF FILE
      WRITE ’Max. record number reached, last record is’
      ESCAPE BOTTOM
    END-ENDFILE
  END-FOR
  I := I - 1
  WRITE ’Record’ I ’:’ W-DAT
  CLOSE PC FILE 7                             /* Close PC file 7
END-REPEAT
END

When you run the program, a window appears in which you specify the name of the PC file from which the data are
to be uploaded. The data are then uploaded from the PC. At the end of each loop, the PC file is closed. 

DISPLAY
This statement is used to specify the fields to be output on a report in column format. A column is created for each
field and a field header is placed over the column.

The notation (rep) may be used to define the number of the printer file where the report is to be output. If this printer
file is defined to Natural as PC, the report will be downloaded to the PC.

For detailed information on this statement, see the Natural Statements documentation.

Example

The following program demonstrates the use of the WRITE and DISPLAY statements for downloading reports to the 
PC.

/* DISPLEX: Example for DISPLAY to PC
/*
DEFINE DATA LOCAL
  01 PERS VIEW OF EMPLOYEES
     02 PERSONNEL-ID
     02 NAME
     02 CITY
END-DEFINE
*
FIND PERS WITH CITY = ’NEW YORK’               /* Data selection
   WRITE (7) TITLE LEFT ’List of employees in New York’ /
   DISPLAY (7)          /* (7) designate the output file (here the PC).
     ’Location’  CITY
     ’Surname’   NAME
     ’ID’        PERSONNEL-ID
END-FIND
END

When you run the program, a window appears in which you specify the name of the PC file into which the report is
to be downloaded. The report is then downloaded to the PC.

  List of employees in New York
  
        Location             Surname           ID
  -------------------- -------------------- ---------
  
  NEW YORK             RUBIN                20007100
  NEW YORK             WALLACE              20025400

DOWNLOAD PC FILE

Operand Possible Structure Possible Formats Referencing Permitted Dynamic Definition 

Operand1 C S A G  A N P I F B D T L C yes no 

Operand2 C S    A          yes yes 

Note: 
Format C is not valid for Natural Connection. It will be rejected at runtime. 

Function

This statement is used to transfer data from the mainframe to the PC.

work-file-number 

The work file number to be used. This number must correspond to one of the work file numbers for the PC as
defined to Natural.

VARIABLE 

The records in the PC file will be written in variable format. Note that variable records cannot be converted to PC
spreadsheet formats.

Field Specification - operand1 

With operand1 you specify the fields to be downloaded to the PC.

COMMAND 

With the COMMAND clause you can send PC commands (i.e any command that can be entered in the command line
of Entire Connection on the PC) from the mainframe to the PC.

Entire Connection checks whether the command sent is valid or not. A command that cannot be recognized by the
PC is rejected. In this case, Natural issues the error message that the downloaded command was rejected by the PC.

You can use the COMMAND clause, for example, to execute Entire Connection tasks from the mainframe. If you
have the task DIR which lists PC directory information, you can initiate this directly out of your Natural program on
the mainframe with the following statement:

DOWNLOAD PC FILE 7 COMMAND ’DIR’

In example 2 below, the COMMAND clause is used to define the name of the PC file that is to receive the
downloaded data. In this way, you can avoid prompting for the name of the file.

Command Specification - operand2 

With operand2 you specify the DOS command or Entire Connection task that is to be executed on the PC.

Operand2 must be an alphanumeric constant or variable.

Related Statements 

CLOSE PC FILE, UPLOAD PC FILE.

Example 1

The following program demonstrates the use of the DOWNLOAD PC FILE statement. The data are first selected and
then downloaded to the PC using work file 7.

/* DOWNLEX1: Example for DOWNLOAD PC FILE
/*
DEFINE DATA LOCAL
  01 PERS VIEW OF EMPLOYEES
     02 PERSONNEL-ID
     02 NAME
     02 CITY
END-DEFINE
*
FIND PERS WITH CITY = ’NEW YORK’               /* Data selection
   DOWNLOAD PC FILE 7 CITY NAME PERSONNEL-ID   /* Data download
END-FIND
END

When you run the program, a window appears in which you specify the name of the PC file into which the data are
to be downloaded. The data are then downloaded to the PC.

         CITY                 NAME         PERSONNEL
                                               ID
  -------------------- -------------------- ---------
  
  NEW YORK             RUBIN                20007100
  NEW YORK             WALLACE              20025400

Example 2

The following program demonstrates the use of the COMMAND clause in the DOWNLOAD PC FILE statement.
The name of the receiving PC file is first defined. Then the data are selected and downloaded to this file.

/* DOWNLEX2: Example for DOWNLOAD PC FILE
/*
DEFINE DATA LOCAL
  01 PERS VIEW OF EMPLOYEES
     02 PERSONNEL-ID
     02 NAME
     02 CITY
  01 CMD (A80)                                 /* Variable for transfer
END-DEFINE                                     /* of the PC command
*
MOVE ’SET PCFILE 7 DOWN DATA PERS.NCD’ TO CMD  /* PC command to define
*
DOWNLOAD PC FILE 6 COMMAND CMD                 /* Command download
*
FIND PERS WITH CITY = ’NEW YORK’               /* Data selection
   DOWNLOAD PC FILE 7 CITY NAME PERSONNEL-ID   /* Data download
END-FIND
END

Note: 
The PC file number in two successive DOWNLOAD PC FILE statements must be different.

When you run the program, the data are downloaded to the PC file that was specified in the program. A window does
not appear.

         CITY                 NAME         PERSONNEL
                                               ID
  -------------------- -------------------- ---------
  
  NEW YORK             RUBIN                20007100
  NEW YORK             WALLACE              20025400

PRINT
This statement is used to produce output in free format.

The notation (rep) may be used to define the number of the printer file where the report is to be output. If this printer
file is defined to Natural as PC, the report will be downloaded to the PC.

For detailed information on this statement, see the Natural Statements documentation.

Example

The following program demonstrates the use of the PRINT statement for downloading reports to the PC.

/* PRINTEX: Example for PRINT to PC
/*
DEFINE DATA LOCAL
  01 PERS VIEW OF EMPLOYEES
     02 PERSONNEL-ID
     02 NAME
     02 CITY
END-DEFINE
*
FIND PERS WITH CITY = ’NEW YORK’                 /* Data selection
   PRINT (7) 5T CITY 20T NAME 40T PERSONNEL-ID    /* (7) designate
                                                 /* the output file
                                                 /* (here the PC).
END-FIND
END

When you run the program, a window appears in which you specify the name of the PC file into which the report is
to be downloaded. The report is then downloaded to the PC.

  NEW YORK             RUBIN                20007100
  NEW YORK             WALLACE              20025400

UPLOAD PC FILE

Structured Mode Syntax

Reporting Mode Syntax

Operand Possible Structure Possible Formats Referencing Permitted Dynamic Definition 

Operand1  S A G  A N P I F B D T L C yes yes 

Operand2  S A G  A N P I F B D T L C yes yes 

Operand3  S       I       yes yes 

Note: 
Format C is not valid for Natural Connection. It will be rejected at runtime. 

Function 

This statement is used to transfer data from the PC to the mainframe.

Note:
No I/O statement may be placed with the UPLOAD PC FILE processing.

work-file-number

The number of the work file to be used. This number must correspond to one of the work file numbers for the PC as
defined to Natural.

Field Specification - operand1-2 

With operand1 and operand2 you specify the fields to be uploaded from the PC. The fields may be database fields or
user-defined variables.

Options 

See the READ WORK FILE statement in the Natural Statements documentation for a description of the ONCE,
SELECT, GIVING LENGTH options.

The RECORD option is not permitted for PC work files. It will be rejected at runtime.

When uploading data, if you wish to define a filler, you must use a dummy variable instead of the standard filler 
notation.

Related Statements 

CLOSE PC FILE, DOWNLOAD PC FILE.

Example

The following program demonstrates the use of the UPLOAD PC statement. The data are first uploaded from the PC
and then processed on the mainframe.

/* UPLDEX: Example for UPLOAD PC FILE
/*
DEFINE DATA LOCAL
  01 EMPL VIEW OF EMPLOYEES
     02  PERSONNEL-ID
     02  INCOME
         03  SALARY (1)
  01 #PID (A8)                                   /* Personnel ID on PC
  01 #NEW-INCREASE (N4)                          /* Increase for salary
END-DEFINE
*
UPLOAD PC FILE 7 #PID #NEW-INCREASE              /* Data upload
*
  FIND EMPL WITH PERSONNEL-ID = #PID             /* Data selection
     ADD #NEW-INCREASE TO SALARY (1)             /* Data update on host
     UPDATE
     END TRANSACTION
     ESCAPE BOTTOM
  END-FIND
*
END-WORK
END

When you run the program, a window appears in which you specify the name of the PC file from which the data are
to be uploaded. The data are then uploaded from the PC.

WRITE
This statement is used to produce output in free format.

The notation (rep) may be used to define the number of the printer file where the report is to be output. If this printer
file is defined to Natural as PC, the report will be downloaded to the PC.

For detailed information on this statement, see the Natural Statements documentation.

Example

The following program demonstrates the use of the WRITE and DISPLAY statements for downloading reports to the 
PC.

/* WRITEEX: Example for WRITE to PC
/*
DEFINE DATA LOCAL
  01 PERS VIEW OF EMPLOYEES
     02 PERSONNEL-ID
     02 NAME
     02 CITY
END-DEFINE
*
FIND PERS WITH CITY = ’NEW YORK’               /* Data selection
   WRITE (7) TITLE LEFT ’List of employees in New York’ /
   DISPLAY (7)          /* (7) designate the output file (here the PC).
     ’Location’  CITY
     ’Surname’   NAME
     ’ID’        PERSONNEL-ID
END-FIND
END

When you run the program, a window appears in which you specify the name of the PC file into which the report is
to be downloaded. The report is then downloaded to the PC.

  List of employees in New York
  
        Location             Surname           ID
  -------------------- -------------------- ---------
  
  NEW YORK             RUBIN                20007100
  NEW YORK             WALLACE              20025400

